Průběh funkce pomocí systému MAPLE.

Vyšetřování průběhu funkce je komplexní a někdy velmi obtížná úloha. V konkrétních aplikacích nás většinou zajímají jen některé otázky týkající se průběhu dané funkce. Například: jaký je počet a přibližná hodnota nulových bodů funkce, nebo jaká je největší hodnota funkce. I v těchto případech je dobré znát o dané funkci maximální počet informací. Počítačový systém MAPLE umožňuje podrobně vyšetřit i takové funkce, pro které analytické metody vyšetřování selhávají. Na druhou stranu bez analýzy funkce bychom "pomocí" Maplu mohli dojít ke zcela chybným závěrům.
Obecně průběh funkce vyšetřujeme v následujících bodech:
(1) $D(f)$, sudost, lichost, periodicita, průsečíky se souřadnicovými osami,
(2) limity v krajních bodech intervalů, které tvoří D(f), spojitost,
(3) monotónnost, extrémy,
(4) intervaly konkavity a konvexity, inflexní body,
(5) asymptoty,
(6) graf funkce.

Následující poznámky a příklady mohou sloužit jako návod, jak lze konkrétně použít systému Maple v jednotlivých bodech vyšetřování funkce.

(1) $D(f)$. **Definiční obor** funkce je většinou (mimo funkce periodické) sjednocení konečného počtu intervalů. Pro stanovení definičního oboru funkce je potřeba vyšetřit, z jakých elementárních funkcí se funkce skládá, a použít vět o definičním oboru součtu funkcí nebo složené funkce.

Při zjišťování definičního oboru funkce může v mnoha případech pomoci v Maplu příkaz *plot* pro zobrazení grafu funkce. Mohlo by se zdát, že takto dostaneme všechny potřebné informace o průběhu funkce a netřeba vyšetřovat další body (2) až (6). U některých funkcí však ze zobrazeného grafu nelze vyčíst pouhým okem ani definiční obor, natož další potřebné vlastnosti funkce. Například funkce $f(x) = \frac{x^2+x-2}{x-1}$ a $g(x) = x+2$ se Maplem zobrazí jako stejné funkce, ale $D(f) = R - \{1\}$ a $D(g) = R$.

Pro zjištění $D(f)$ v příkazu *plot* volíme "dostatečně velký" zobrazovací interval funkční proměnné. To však může způsobit, že se ztratí detaily funkce okolo tzv. kritických bodů (např. kde je první nebo druhá derivace rovna 0 nebo tam, kde derivace neexistuje). V příkladu 1 se při zobrazení grafu na větším intervalu nedá zjistit chování funkce v okolí bodu 0, nejsou vidět inflexni body (nacházejí se v bodech -1 a 1) a není jasné, zda má funkce asymptoty (nemá).

Příklad 1. Funkce $f(x) = \ln(1+x^2)$

```
> f1:= x -> ln(1+x^2);
```

$$f1 := x \to \ln(x^2+1)$$

```
> plot(f1(x), x=-70 - .5..70 + .5, color = red);
```



Naopak, pokud není vyšetřen celý definiční obor funkce, potom při zobrazení grafu funkce v Maplu může dojít k přehlédnutí podstatné části funkce. V příkazu *plot* lze totiž snadno stanovit příliš malý zobrazovací interval a tím se omylem omezit jen na jednu z větví grafu viz následující příklad.

Příklad 2. Funkce

$$f(x) = \frac{x^2}{2x+5}$$

```
> f2:= x -> x^2 / (2 * x + 5);
```

$$f2 := x \to \frac{x^2}{2\,x+5}$$

```
> plot(f2(x), x = -2..4, color = red,discont = true);
```



ale $D(f2) = \mathbb{R} - \{-5/2\}$ a graf funkce $f2$ má dvě větve.

```
> plot(f2(x), x = -5..5,y = -30..30, color = red,discont = true);
```



Při určování definičních oborů funkcí využijeme znalosti definičních oborů elementárních funkcí. Tak třeba v příkladu 1 je funkce ln definována jen pro kladný argument, tj. řešili bychom

```
> solve(x^2+1>0,x);
```

což by bylo zbytečné, neboť jednoduchou analýzou je ihned patrné, že $D(f1) = \mathbb{R}$. U příkladu 2 použijeme příkaz Maplu :

```
> solve(denom(f2(x))=0,x);
```

$$\frac{-5}{2}$$

Tímto příkazem nalezneme hodnoty, ve kterých funkce není definována, tedy $D(f2) = \mathbb{R} - \{-5/2\}$.

Sudost nebo **lichost** funkce dokážeme pomocí příkazu,

```
> f(-x);
```

Obdobně, pokud chceme najít **průsečíky se souřadnicovými osami**, použijeme příkazy Maplu. Průsečík s osou x je možno analyticky nalézt jenom výjimečně, systém Maple použije Newtonovy metody a většinou najde řešení. Konkrétně pro funkci z příkladu 1 dostáváme:

```
> f1(-x);
```

$$\ln(1+x^2)$$

Tedy f1 je sudá funkce.

```
> f1(0);
```

$$0$$

Našel se průsečík s osou y, $y=0$.

```
> solve(f1(x)=0,x);
```

$$\{0\}$$

Našel se průsečík s osou x, $x=0$.

(2) Limity v krajních bodech intervalů, které tvoří D(f) , nám dávají odpověď i na otázky spojitosti funkce. **Spojitost** vyšetřujeme tak, že použijeme vět o spojitosti součtu, součinu a podílu elementárních funkcí a věty o spojitosti složené funkce. Ve zbylých "nepříjemných bodech", do kterých patří krajní body definičního oboru, počítáme limitu a zjišťujeme, zda se rovná funkční hodnotě. Pokud limita neexistuje, vyšetřujeme jednostranné limity nebo dokážeme jejich neexistenci.

Vyšetřování prováděné v bodě (2) demonstrujeme na příkladu 2 (příklad 1 je v tomto smyslu jednoduchý, limity vyšetřete sami). Je tedy nutné vypočítat následující limity v krajních bodech intervalů $D(f)$:

$$\lim_{x\to\infty} f2(x)\,, \qquad \lim_{x\to-\infty} f2(x)\,, \qquad \lim_{x\to-5/2\pm} f2(x)\,.$$

```
> limit(f2(x), x = infinity);
```

$$\infty$$

```
> limit(f2(x), x = -infinity);
```

$$-\infty$$

```
> limit(f2(x), x = -5/2);
```

$$undefined$$

```
> limit(f2(x), x = -5/2,right);
```

$$\infty$$

```
> limit(f2(x), x = -5/2,left);
```

$$-\infty$$

Tedy $\lim\limits_{x \to -5/2} f2(x)$ neexistuje, existují pouze nevlastní jednostranné limity. Limitu $\lim\limits_{x \to -\infty} f2(x) = -\infty$, kterou jsme právě vypočítali, bychom z grafu funkce $f2$ omezeného na interval $x \in < -5\,, 5 >$ (viz výše) nedokázali vyčíst.

(3) Monotónnost, extrémy. Jak jsme uvedli v bodě (2), definiční obory běžných funkcí se skládají z konečného počtu intervalů, na kterých je funkce spojitá. Stejně tak lze $D(f)$ rozložit dále na konečný počet intervalů, na kterých je funkce monotonní. Pokud tyto intervaly určíme a vypočteme-li limity v krajních bodech těchto intervalů, jsme schopni odpovědět na to, kde se nacházejí lokální (i absolutní) extrémy funkce, a stanovit **obor hodnot funkce H(f)**. Monotonii funkce vyšetřujeme standardně pomocí první derivace funkce. Občas jsme schopni okamžitě zjistit monotonii funkce bez užití derivace, pokud je funkce složením dvou nebo více monotonních funkcí. Tak je tomu u následujícího příkladu.

Příklad 3.

$$f(x) = \frac{1}{1+\mathrm{e}^{\frac{1}{x}}}\,,$$

je funkcí rostoucí na intervalu $x \in (-\infty\,, 0)$ i na intervalu $x \in (0, \infty\,)$,, (dokažte), ale není rostoucí na sjednocení těchto intervalů, tj. na celém D(f). Graf znázorníme Maplem .

```
> f3:= x -> 1/(1+exp(1/x));
```

$$f3 := x \to \frac{1}{1+e^{\left(\frac{1}{x}\right)}}$$

```
> plot(f3(x), x=-10- .5..10 + .5, color = red, discont = true);
```



Nyní vyšetříme monotonnost a extrémy funkce z příkladu 1.

```
> df1:=D(f1);
```

$$df1 := x \to \frac{2\,x}{1+x^2}$$

```
> solve(df1(x)=0,x);
```

$$0$$

Bod 0 je tzv. stacionární bod a může v něm být extrém. Je v něm lokální (a zároveň absolutní) minimum funkce, neboť:

```
> solve(df1(x)<0,x);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(0))$$

```
> solve(df1(x)>0,x);
```

$$\mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

V bodě (4) ukážeme, že je druhá derivace funkce f1 v bodě 0 kladná. Z vyšetřování v bodě (2) plyne, že funkce $f1$ nenabývá svého absolutního maxima.

Někdy nelze derivaci v některých bodech počítat jen formálně. Například tam, kde nemůžeme použít větu o derivaci složené funkce pro funkci arccos , protože tato funkce nemá v bodech $-1, 1$ derivaci. Následuje takový obtížnější příklad a Maple nám ho pomůže vyřešit.

Příklad 4. Funkce

$$f(x) = \arccos^2\left(\frac{1}{1+x^2}\right)$$

```
> f4:=x->(arccos(1/(1+x^2)))^2;
```

$$f4 := x \to \arccos(\frac{1}{1+x^2})^2$$

```
> solve((1/(1+x^2))>=-1 and 1/(1+x^2)<=1,x);
```

$$x$$

Tedy $D(f4) = \mathbb{R}$. Formální derivování (pozor na stejné značení jako definiční obor) dává:

```
> df4:=D(f4);
```

$$df4 := x \to \frac{4 \arccos(\frac{1}{1+x^2})\, x}{(1+x^2)^2 \sqrt{1 - \frac{1}{(1+x^2)^2}}}$$

Pravá strana není definována pro $x = 0$, ale z toho neplyne, že $f'(0)$ neexistuje.

```
> df4(0);
Error, (in df4) numeric exception: division by zero
```

```
> limit(f4(x),x=0);
```

$$0$$

```
> limit(df4(x),x=0);
```

$$0$$

Funkce $f4$ je spojitá v 0 a $\lim_{x\to 0} f4'(x) = 0$, tedy můžeme usoudit, že $f'(0) = 0$. Pro ověření nakresleme graf funkce $f4$ (červená čára) spolu s grafem derivace funkce (zelená čára).

```
> plot([f4(x),df4(x)],x=-4..4,discont=true,thickness=[2,1]);
```



(4) Intervaly konvexnosti resp. konkávnosti, inflexní body. Nejčastěji vyšetřujeme konvexnost a konkávnost zadané funkce pomocí druhé derivace. Tak jako v předešlých bodech je nutné znát matematické věty, o které se můžeme při tomto vyšetřování opřít.

Vrátime se znovu k příkladu 1 a na základě znalosti první derivace vypočítáme derivaci druhou.

```
> df1:=D(f1);
```

$$df1 := x \to \frac{2\,x}{1+x^2}$$

```
> ddf1:=D(df1);
```

$$ddf1 := x \to \frac{2}{1+x^2} - \frac{4\,x^2}{(1+x^2)^2}$$

```
> solve(ddf1(x)<0,x);
```

$$\mathrm{RealRange}(-\infty,\ \mathrm{Open}(-1)),\ \mathrm{RealRange}(\mathrm{Open}(1),\ \infty)$$

Zřejmě v uvedených intervalech je funkce konkávní, jinde konvexní. Druhá derivace funkce $f1$ v bodě 0 je kladná; to je potvrzení toho, že funkce má v bodě 0 lokální minimum. Skutečně:

```
> solve(ddf1(x)=0,x);
```

$$-1,\ 1$$

v bodech [-1 , ln(2)], [1, ln(2)] jsou inflexní body grafu funkce $f1$:

```
> f1(1);
```

$$\ln(2)$$

5. Asymptoty. Hledají se přímky $y = kx + q$, které by byly šikmými asymptotami k zadané funkci, nebo svislé asymptoty, tj. přímky kolmé k ose x tvaru $x = c$ v bodech c, kde existuje alespoň jedna jednostranná nevlastní limita funkce. Tedy hledání asymptot obou typů se převádí na počítání limit.

V příkladu 1 nemá funkce asymptoty (ověření proveďte sami). Výpočet asymptot budeme demonstrovat na příkladu 2.

Již v bodě (2) jsme u funkce $f2$ zjistili, že v bodě $-5/2$ existují nevlastní jednostranné limity $\lim\limits_{x\to -5/2-} f2(x) = -\infty$ a $\lim\limits_{x\to -5/2+} f2(x) = -\infty$, tedy $x = -5/2$ je jediná svislá asymptota funkce.
Někdy je možno zjistit šikmé asymtoty přepsáním funkce do jiného tvaru pomocí následujícího příkazu Maplu:

```
> convert(x^2 / (2 * x + 5),parfrac,x);
```

$$\frac{x}{2} - \frac{5}{4} + \frac{25}{4\,(2\,x+5)}$$

tedy lineární funkce $y = (1/2) * x - 5/4$ je šikmou asymptotou funkce.
Ověříme šikmou asymptotu výpočtem příslušných limit.

```
> k:=limit(f2(x)/x,x=-infinity); q:=limit(f2(x)-k*x,x=-infinity);
```

$$k := \frac{1}{2}$$

$$q := \frac{-5}{4}$$

6. Graf funkce. O různých možnostech zobrazení grafu funkce v Maplu pojednává část **Graf funkce jedné proměnné**. Omezíme se zde proto na znovu zobrazení funkce z příkladu 1 v intervalu, kde již jsou patrné její detaily,

```
> plot(f1(x), x=-2 - .5..2 + .5, color = red);
```



a na zobrazení funkce z příkladu 2 spolu s její šikmou asymptotou v $\pm\infty$.

```
> y:= x -> x/2 - 5/4;
```

$$y := x \to \frac{1}{2}x - \frac{5}{4}$$

```
> plot([f2(x),y(x)],x=-10..10,y=-10..10,discont=true,thickness=[2,1]);
```



Z grafu funkce $f1$ a z předešlého vyšetřování v bodě (3) je patrné, že $H(f1) =< 0, \infty)$ (obor hodnot $H(f2)$ vyšetřete sami).